
Starting a Session in the Redatam+SP 
Process Module 
 

I. The Database Dictionary  
 
The database dictionary (.dic file) is the main object from where the user can manipulate, analyze and 
browse the data contained in a Redatam database. The first thing to do  in a session is to open a dictionary 
file from de main menu File>Open>Dictionary. 
 
The right side of the dictionary window displays the hierarchical database structure as a directory-like tree of 
entities, each of which has individual elements or members, which are often geographical, like provinces, 
districts, or city blocks, but can also be individual persons, houses, schools, etc. The left-side box contains 
the list of variables describing the currently high-lighted entity on the right side. Click on another entity name 
to change the current entity. The list of variables in the left box is then changed accordingly. The 
presentation of the dictionary can be inverted horizontally or vertically. 
 
Double click on a variable name (on the left side list) to open a window displaying the information about the 
variable. Double click on an entity name and the branches beneath this entity will collapsed or expanded. 
 
 

II. Programming: generating outputs 
 

THE BASIC COMMAND SET 
 
R+SP has three levels of programming to get results from a database: a) Fast Results; b) Assisted 
Programming; and c) Free Editor.  The first level is designed for you to get results using only the mouse, in 
a "point and click" fashion, requiring no programming skills at all. However, this level has some limitations, 
as not to be able to build derived variables. The second level is based on a Programming Assistant, which 
needs very little knowledge of the R+SP language, and builds full programs that use all the system’s 
flexibility and power. The third level is for the programmers who are already familiar with the R+SP 
language, and will be more at home with this totally free format of programming. This level also provides an 
assisted wizard to verify the syntax and to write blocks of commands. 
 
Each tool  (Easy Frequency, Easy Crosstabs, etc.) is in fact a visual way to create a program in the 
REDATAM+SP language. The same result can be obtained through programming a command set.  In a 
command set, the user tells REDATAM+SP to carry out processes to create new variables and produce 
specific tabulations and other results via a set of commands from the REDATAM+SP language.  
 
The programs are stored in files with the extension .spc (Statistical Processor Commands). To create a new 
program, or open an existing program, choose the function File in the main menu.  

Then you can select New for a new object, or Open an existing object. In any case (New or Open), choose 
Command Set, which has the options of Free Format and Assisted. In the case of an existing program the 
system opens a dialog box to navigate and bring the desired program. In the case of a new program the 
system opens directly the window of free editing or the assisted window, according to the choice.  
To open a program, it is necessary to have an active database; otherwise the File menu will not habilitate 
the Command Set option. In that case it is only possible to open a dictionary relative to a database. 
 
There are two ways of programming in R+SP:  
a) using the Programming Assistant; or 
 b) writing the program directly in a text Free Format Editor.  
 
These two ways mix themselves, and not only the Programming Assistant has a free window to enter the 
commands, but the Free Format Editor as well has an assistant (identical to the Programming Assistant) to 
aid in the writing of the commands.  
 
A R+SP program can be created in one way (Assistant) and later worked on in another way (Free Format 
Editor, for instance).  It is possible to open more than one program window at the same time, them being 
free or assisted windows. 
 
The language has three basic commands: 
 

RUNDEF to define the environment in which a process will occur, including the specific selection set 
to be used or/and the universe. 

DEFINE to create new variables and indicate their nature and scope. 
TABLE to indicate what the specific process and output should be. 

 
Each of these main instructions can be qualified by subordinate clauses. For example, to DEFINE a new 
variable, the user can specify its nature via clauses including AS, TYPE, RANGE, VARLABEL, 
VALUELABEL, limit the cases to which it applies by a FOR clause (which is similar to an “if” in Redatam-
Plus), and RECODE the values, QUANTIFY and COUNT elements, etc.  
 
DEFINE hhold.newvar 
AS COUNT person 
TYPE INTEGER 
VARLABEL “Total persons in the house” 
 
Similarly, the output can be specified with the TABLE command using AS <tabletype>, where tabletype may 
be FREQUENCY, AVERAGE, CROSSTABS, AREALIST, etc., OF a set of variables, and further specified 
by clauses beginning with FOR, etc. 
 
TABLE tab1 
AS FREQUENCY OF 
Hhold.newvar 

Scheme 

 
A simple Program using RUNDEF and TABLE  

Write a program to get the same results of frequencies, crosstabs and averages that are produced by 
the Wizards.  
 
Procedure  
 
1 Opening the Command Editor: To open the Command Editor screen (free format), go to the main 

menu, select File, and then choose New >Program >Free format.  
 

2 Save As: It is recommended to do a Save from time to time, before executing the program and 
when you finish your job. Go to the main menu, choose File >Save and give the program a 
meaningful name, such as Prog1Freq. You can use also the Save icon in the Toolbar. 

 
3 RUNDEF, always the first command: A Program must start with a RUNDEF, specifying the 

Selection and other general filters, if needed. It is always the very first command, and it must be 
unique in the program. Redatam makes your job easier by writing a default RUNDEF command 
every time you start a new program. You can use the Command Assistants (there are the 
RUNDEF Assist, the DEFINE Assist and the TABLE Assist). 

 
4 Accessing the RUNDEF Assist: After opening the Free Editor, you can use the specific Assistants 

by means of the popup menu, with the right mouse button. Block the whole RUNDEF command by 
positioning the mouse before the R of RUNDEF, and holding the mouse left button, move over the 
two lines of the command until the end of the word ALL, and then press the mouse right button. 

 
5 Fill in the spaces in the RUNDEF Assistant: 

 
Name, accepts any text. 
Selection, defines the area that you want to process. By default it will read all the database (ALL), 
and the assistant does not show anything in the selection box. If you want to use a specific 
selection, you should have already created the selection file (slw), and use the browse button to 
look for it in the disk. 
Universe, optional. This clause defines the filter expression to select the cases to process.  
Go back to the Command Editor with a click in the OK button in the bottom part of the Assistant 
screen, or use the Save button in the Toolbar. 
 
Specifying the output table  – the TABLE command.  Generally, each result needs one separate 
TABLE command. So, the program will have one TABLE command for each set of frequencies, 
crosstabs and averages (FREQUENCY, CROSSTABS and AVERAGE). 

 
6 Set the table command position in the program (TABLE): Click the mouse in the next free line after 

the RUNDEF command. If there is no free line, click the mouse at the end of the second line of the 
RUNDEF command and press the [ENTER] key, and then click the mouse in the new line. This 
will be the place where you will put the first TABLE command. Once you finish with the TABLE 
Assistant (see next step), the command will be inserted into the Command Editor in the same 
place where the cursor is—if you want, you can leave an empty line after the RUNDEF command 
to improve its readability. 

7 Use the TABLE Assistant: Having the Editor as the active window, that is, its title bar being navy 
blue, call the popup menu (mouse right button) and select the TABLE Assist. You will build the first 
TABLE as a frequency (TABLE AS FREQUENCY). 

 
 

 
Demonstration 

Compute the distribution of dwellings by type and distribution of population by gender. Open new 
commands set and type the following program: 
 

RUNDEF program1 
 SELECTION ALL 
 
TABLE  Freq1 
    AS FREQUENCY  
    OF POP99.AGE99, POP99.SEX99, HOUSE99.SEWER99 

 
Run the program and check the results 
 

 
Notes: 

1. The name of the program “program 1”could be any.  
2. “One” is the name of the table. It could be any user’s defined name. 
3. Both the RUNDEF and TABLE require a name. 
4. The clause AS is followed by the type of table required. 
5. The clause OF is followed by the list of variable identifiers to tabulate. 
6. A variable identifier is always built from the entity name (HOUSE99) followed by a dot immediately 

followed by the variable name (POP99). NO BLANK (space) must appear in the variable 
(POP99.AGE99). 

7. A command set must include one (and only one time) a RUNDEF command, one DEFINE command 
for each derived variable (optionally and could be more than one) and one TABLE command for each 
output required (could be more than one TABLE command). Several table commands may appear in 
the same command set. 

8. The RUNDEF/SELECTION clause allows you to select a different geographical area to process. 
9. Optionally use the Assists windows to build the table command. 
 
 
 
 

 
Exercises 

1. Create a frequency table of the variables tenure, roof, room, and sewer. 
2. Create a frequency of any variable from the household type.  
3. Compute the sex distribution by type of household. 
4. Compute the sex distribution by relation with Household Head 
5. Compute the sex distribution by relation with Household Head and marital status. 
6. Compute the Average age by sex and marital status. Comment the results. 

III. Using filters in the output table: the clause FOR 
 
To restrict the scope of the tabulation command or the definition of a variable we use the clause FOR 
followed by a logical expression.  
 
The result of a logical expression is of BOOLEAN type: TRUE or FALSE. The logical expression is 
evaluated and only those for which the logical expression is TRUE are included in the TABLE. 
 
Building a logical expression requires the use of the logical operators: 
= equal 
> greater than 
< less than 
<> not equal than 
>= greater or equal than 
<= less or equal than 
And the relational operators:  NOT AND OR 
 

TABLE table1 
Syntax 

 AS FREQUENCY 
  OF <variable list> 
 FOR <logical expression> 
TABLE table2 
 AS CROSSTABS 
  OF <var list> BY <var list> [ BY <var list> [ BY  <var list> ] ] 
 FOR <logical expression> 
TABLE table3 
 AS AVERAGE 
  OF <variable > BY <var1> [ BY <var2> ] 
 FOR <logical expression> 

 

Computing the average household head age by sex 
Demonstration 

 
RUNDEF EasyCross 
   SELECTION ALL 
 
TABLE Table1 
   AS AVERAGE 
   OF POP99.AGE99 BY POP99.SEX99 
   FOR POP99.RELATION99 = 1 

 
 

 
Exercises 

1. Compute the average household head age by sex by urban/rural areas. 
2. Compute a Crosstabs between type of energy supply and sewage connection for rural areas. 
3. Compute a Frequency of economic activity last week for female headed households and in the same 

command set compute a Crosstabs between education level and marital status for single female 
headed households. 

IV. Using filters: the clause UNIVERSE 
 
The UNIVERSE clause, which can be used only in the RUNDEF command, restricts the entire run to cases 
that comply with the specified Boolean expression, i.e. when the expression is true. It is use as a global 
filtering. The restriction applies to all the outputs of a Run. 
 

 
Syntax 

RUNDEF <runid> 
 SELECTION <selection> 
 UNIVERSE <logical expression> 
 

 

 
Demonstration 

Compute the sex distribution only of the population that performed Home duties as their economic activity  
during the previous week of the census. 
 

RUNDEF program3 
   SELECTION ALL 
   UNIVERSE  POP99.ECONOACTV99 =3 
 
TABLE one  
  AS FREQUENCY 
  OF POP99.SEX99 

 

 
Notes 

1. The logical expression must be defined only with variables that exist permanently on disk; it cannot 
include DEFINED variables from the current command set.  

2. The Keyword FOR can be used instead of UNIVERSE. 
3. If a DEFINED variable will be saved in the command set the UNIVERSE cannot be applied. 
 
 

 
Exercises 

1. Compute the average age at first child by education level for female headed households only 
2. Compute the average of hours worked last week by economic activity last week for female head of 

households only. 

V. Geographic Area Selection 
 
Since the user seldom wants to process the entire data set with REDATAM+SP, the user can identify a 
selection set that defines the specific elements (usually geographic areas) that will be processed to obtain 
statistical results. Any number of different selection sets may be defined for a given REDATAM+SP 
database and each selection set can be built from any combination of areas to define the final selection.  
 
The area selection provides REDATAM+SP a mean to target on specific area of the database. Traditionally, 
the selection of a sub database was related to the geography, for this reason the sub database is selection 
is called area selection. 
 
 

 
Steps to create a selection 

 Open the selection window by choosing File >New >Selection in the main menu. 
 The system displays the selection screen with the database structure, showing the root entity, that 

is, the most aggregated level in the database. 
 Use the expand option, either from the popup menu or from the toolbar that is also shown. The 

hierarchical structure "opens" in its second level. 
 
 
  
The toolbar buttons are the same as the options that appear in the popup menu. 
 
The elements in this second level (generally they correspond to geographical entities) are displayed with 
their identification codes, their element names, or both. 
 
 Click in any element and repeat the process to expand it into its lower elements. 
 Locate the cursor over the element you want to select. 
 Mark the element with a double click, or using the select option in the popup menu, or use the 

toolbar to select the element. The small square at the left side of the element should become 
yellow.  

 Repeat the selection process for all those elements you want to include in the new selection set. 
 
 If you select an element that was not expanded into its lower elements (does not have the  symbol 

on its side), all the lower elements under this level will be selected. 
 To contract an expanded branch with all its elements, position the mouse in the corner of the 

element in the tree that has the symbol and click it. 
 To save the selection set, go to the main menu and choose the File >Save option. Use the save 

dialog box to find the directory where you want to save the file and give it a name (.slw extension 
by default).  

 A double click on a selected element deselects it. 

QUIZ 1 
 
Name: ________________________________________ 
 
Date: _________________________________________ 
 
 
Please answer the following questions 
 
1. What is the average number of children in female between 25 and 35 years old? 
__________________________________________________________________ 
__________________________________________________________________ 
__________________________________________________________________ 
 
2.  Calculate the average age of head of households female 
__________________________________________________________________ 
__________________________________________________________________ 
__________________________________________________________________ 
 
3. What is the distribution of population attending school? 
__________________________________________________________________ 
__________________________________________________________________ 
__________________________________________________________________ 
 
4. What is the range defined for the variable Type of Dwelling? 
_________________________________________________________________ 
__________________________________________________________________ 
__________________________________________________________________ 
 
5. How many branches does this database contain? 
__________________________________________________________________ 
__________________________________________________________________ 
__________________________________________________________________ 
 
6.  How many dwellings don’t have piped water in their houses? 
_______________________________________________________________ 
_______________________________________________________________ 
_______________________________________________________________ 
 
7.   Find out the number of males headed households of 45 years old. 
______________________________________________________________________________________
______________________________________________ 

QUIZ 2 
 
Name: ________________________________________ 
 
Date: _________________________________________ 
 
 
Please answer the following questions 
 
1. Compute the distribution by education level by age group for female headed households over 15. 
 
2. Compute the distribution of school attendance by age for women between 6 and 18. 
 
3. Where do you find a greater concentration of students? 
__________________________________________________________________ 
__________________________________________________________________ 
__________________________________________________________________ 
 
4. Which is the difference between the UNIVERSE and FOR clauses? 
__________________________________________________________________ 
__________________________________________________________________ 
__________________________________________________________________ 
 
5. How should it be a command set that computes the distribution of mothers by age group? 
 
__________________________________________________________________ 
__________________________________________________________________ 
__________________________________________________________________ 
__________________________________________________________________ 
__________________________________________________________________ 
__________________________________________________________________ 
__________________________________________________________________ 
 
8. What is the distribution of economic activity of single female-headed households?  

VI. Deriving new variables – The RECODE Command 
 
 
The DEFINE command is used to derive new information from the database. This is stored in a temporary 
variable and its definition involves various types of operations such as summing, recodification, arithmetic or 
logical expression, reading data from external source etc.  
 
The RECODE command is used to create a new variable based on the recoding of an existing one. 
 
 

 
Syntax 

 DEFINE entity.newvar 
  AS RECODE <variable>  < recoding scheme> 
  TYPE INTEGER 
  RANGE min - max 
 
 

 
Recode scheme format                Example 

• (<previous value> = <new value>)       (3 = 1) 
 
• (< previous value1> – <previous value2> = <new value>)  (2 – 5 = 3) 
 
• (<previous value1> TO <previous value2> = <new value>)  (6 TO 10 = 4) 
 
• (LOWEST TO <previous value> = <new value>)     (LOWEST TO 4 = 2) 
 
• (<previous value> TO HIGHEST = <new value>)    (11 TO HIGHEST = 5) 
 
• ELSE newvalue      (12-20 = 12) ELSE 13 
 

 
Notes 

1. Values not included in the recoding scheme keep their original value 
2. The ELSE keyword re-assigns all ranges not previously assigned 
3. Variable type is always INTEGER 
4. The RANGE value must fit with the maximum & minimum of the recoding scheme 
 
 

 
Demonstration 

Create a new variable by assigning each person to a group according to its age. The fist group being the 
children, the second the adults and the third the elders. The age range is 0 to 14 for children, 15 to 64 for 
adults and 64 or more for the last group. 

Expressed in natural language, the predicate for realising this operation would be: 

The translation of this predicate into a more REDATAM+SP structure would be
Create a new individual variable as 1 for persons from 0 to 14, 2 for persons 15 to 64 and 3 for persons 64 
and over. 

: 

 
The translation of this predicate into the exact REDATAM+SP  syntax
DEFINE POP99.AGEREC3 

: 

 AS RECODE   POP99.AGE99  
              (0 - 14=1)(15 - 64=2)(64 - Highest = 3)  

TYPE INTEGER 
RANGE 1 - 3 

 
NOTE: 
1. The variable type must be included in the definition. In this particular case, the type of this new variable is 
obviously an integer variable. 
2. The range must be also included in the definition. Since, the variable can take only 3 values; the range is 
1 to 3 
 
 
 
 

 
Exercise 

1. Recode age into groups of five years each. 
2. Recode households into those with and without piped water. 
3. Recode households into those with and without electricity. 
4. Recode households into those with sewer connection and without. 

VII. Promoting information - The COUNT command 
 
 
The DEFINE...COUNT command returns the actual number of elements belonging to an entity. For example 
the number of persons in a household is computed: 
  
  household n  (owner entity) 
   person 1  (counter entity) 
   person 2 
   person 3 
   person 4 
 
The resulting variable contains 4. 
 

 
Syntax 

DEFINE owner.totcount 
  AS COUNT lowerent 
  TYPE INTEGER 
 
where owner is the owner entity, totcount the resulting variable and lowerent the counted entity. 
 
 

 
Notes 

1. The “counted” entity may be of any level below the owner entity. It is then possible to count the total 
number of persons by household or any geographic area to which belong the entity POP99. 

2. The “counted” entity must belong to the same branch of the database. 
3. The TYPE of the resulting variable is always an INTEGER variable. 
4. To evaluate the resulting value the RANGE..CODEBOOK can be used.  
5. A “counted” variable is always a post-process variable 
 
 
 
 
 

 
Exercises 

1. Find out the largest population by District in the country 
2. Find the distribution by sex of elders. 

VIII. Promoting information - The SUM function and The OPTIONS 
WEIGHT function 

 
 
Is used to sum the values of a variable at a lower entity level to a new variable defined at a higher entity 
level, taking into account only elements in the Selection set and only those that pass the DEFINE…FOR or 
RUNDEF…UNIVERSE filters, if any. 
 

 
Demonstration 

DEFINE  HOUSE99.SUMAGE 
   AS SUM   POP99.AGE99 
   TYPE INTEGER 
    

The same result can be obtain using the OPTIONS WEIGHT function 
 

DEFINE HOUSE99.SUMAGE 
AS COUNT PERSON 
OPTIONS WEIGHT  POP99.AGE99 

 
 
The following command set produces the same result: 
 
RUNDEF job 
   SELECTION ALL 
    
 DEFINE HOUSE99.HAGE 
 AS SUM POP99.AGE99 
 TYPE INTEGER 
  
  TABLE TT 
 AS FREQUENCY OF HOUSE99.HAGE 
  
 DEFINE HOUSE99.HAGE2 
 AS COUNT POP99 
 OPTIONS WEIGHT POP99.AGE99 
 TYPE INTEGER 
 
 TABLE TTT 
 AS FREQUENCY OF HOUSE99.HAGE2 
 
 

1. If there are three persons in a house, aged 25, 35, and 42, the value of HOUSE99.SUMAGE for this 
house would be 102. 

Notes 

2. The variable to be summed must normally be quantitative such as age, number of rooms in the 
dwelling, etc. The same cautions involving hierarchical processing apply to SUM. 

3. Do not use parenthesis with the SUM clause, since it gives a compile error. 
 

 
Exercise 

1. Calculate number of households with no water piped for each ed (give the results in  number or 
persons). 

IX. Using the DEFINE/SAVE command 
 .  
When a derived variable is of public general interest or when it is used repeatedly, one should consider 
saving permanently the variable in the database. It then becomes part of the dictionary and can be used as 
an original variable.  
 

 
Syntax 

 RUNDEF saving 
  SELECTION ALL 
 
 DEFINE HOUSE99.HHAGE 
  AS POP99.AGE99 
  FOR POP99.RELATION99 = 1 
  TYPE INTEGER 
  RANGE 0-99 
  VARLABEL “Household Head Age” 
  SAVE HHAGE   OPTIONS OVERWRITE 
 

1. In order to be saved a variable must be computed for the entire database. Thus the only SELECTION 
permitted is ALL (see RUNDEF) 

Notes 

2. The default directory where the bin file is stored is the current working directory. 
3. When the variable must be replaced, it must be first deleted from dictionary and use the OPTION 

OVERWRITE always. 
 
 

 
Demonstration 

Create and save total number of person by household and total number of household by District 
 
   
 
 
 
 
 
 
 

 
Exercise 

1. Compute and save the household sex variable of the head of the household. 
2. Compute and save the total number of persons under 14 years old. 

X. The SWITCH function 
 
The SWITCH clause is very useful to define a new variable depending upon a series of conditions and 
expressions that requires the evaluation of more than one variable, using the form if … then … 
 
As far as the definition of a new variable is concerned, this is the most powerful clause in R+SP, because it 
can use multiple variables and logical expressions to obtain the final result to be assigned to one solely new 
variable. 
 
Examples 
 

a) You have to create a new variable that will classify Households into a new variable 
“Sex_Age_Head” according to the following conditions based on the variables: Sex of Head of 
Household and Age of Head of Household. 

 
   DEFINE HOUSIN.AGE_SEX 
   AS SWITCH 
   INCASE  PERSON.RELAT  = 1 AND    PERSON.SEX   = 1 AND    PERSON.AGE   <= 20 
   ASSIGN 1 
   INCASE  PERSON.RELAT  = 1 AND    PERSON.SEX   = 2 AND    PERSON.AGE   <= 20 
   ASSIGN 2 
   INCASE  PERSON.RELAT  = 1 AND    PERSON.SEX   = 1 AND    PERSON.AGE   >= 21 and PERSON.AGE   <= 45 
   ASSIGN 3 
   INCASE  PERSON.RELAT  = 1 AND    PERSON.SEX   = 2 AND    PERSON.AGE   >= 21 and PERSON.AGE   <= 45 
   ASSIGN 4 
   INCASE  PERSON.RELAT  = 1 AND    PERSON.SEX   = 1 AND    PERSON.AGE   >= 46 
   ASSIGN 5 
   INCASE  PERSON.RELAT  = 1 AND    PERSON.SEX   = 2 AND    PERSON.AGE   >= 46 
   ASSIGN 6 
   RANGE 1 - 6 
   TYPE INTEGER 
   FOR  HOUSIN.TYPHOU <= 8 
   VARLABEL  "Households Head characteristics" 
   VALUELABELS 1  "Male under 20" 
               2 "Female under 20" 
               3 "Male between 21 - 45" 
               4 "Female between 21 and 45" 
               5 "Male over 46" 
               6 "Female over 46"  
 
The Output table gives us the following results: 
 
Frequency
  of Households Head characteristics

Categories Counts % Cumul %
 Male under 20 200            1.8% 1.8%
 Female under 20 61              0.5% 2.3%
 Male between 21 - 45 4,598         40.4% 42.7%
 Female between 21 and 45 888            7.8% 50.5%
 Male over 46 4,177         36.7% 87.2%
 Female over 46 1,451         12.8% 100.0%
 Total 11,375       100.0% 100.0%

NotApp : 2,544          

The value for the new variable Housing.Age_Sex  will depend on the value of the variables Sex and Age: if 
sex = 1 it means it is male and if Age less than 20 it means that falls in the first case evaluated, therefore 
the new variable assumes a value of 1. 
 
Notice that the collective houses are left aside with the expression FOR and only the private Households are 
considered. 
 

b) Another example is the case were you want to define a Household according to its unmeet 
basic needs bsed on accessibility to services such as water, electricity or sewerage. 

 
 

DEFINE HOUSIN.NBI 
AS SWITCH 
INCASE HOUSIN.WATER > 7 
ASSIGN 3 
INCASE HOUSIN.FLOOR > 4 
ASSIGN 3 
INCASE HOUSIN.WATER = 6 
ASSIGN 2 
OPTIONS DEFAULT   1 
TYUPE INTEGER 
RANGE 1 – 3 
 
The value for NBI will be 3 if WATER > 7 or FLOOR > 4. Then the system checks again the WATER 
variable, and if it is equal 6, NBI will be 2, otherwise it will be assigned (OPTIONS DEFAULT  1). Note that 
this condition works like an ELSE for the cases left outside the previous conditions. 
 
Notes 
 

• The SWITCH clause should be used when the new variable depends on more than one source 
variable. 

• It works like an IF … THEN … ELSE, or a CASE structure of the regular programming languages. 
• The testing and assigning conditions work in pairs, a testing condition should always be followed 

by an assigning expression. 
• As soon as a testing condition (INCASE) is True, the system assigns the value calculated by the 

next ASSIGN to the new variable and EXITs the evaluations. This is very important because the 
new value will depend on the order in which the conditions are written. 

XI. Using logical expressions in arithmetic expression 
 
A logical expression is generally used in the context of a filter, i.e. a clause FOR but it can also be used as a 
condition to define a new variable under the AS clause. The result of a logical expression is of BOOLEAN 
type: TRUE or FALSE. The REDATAM+SP language extend the BOOLEAN type to INTEGER so that 
arithmetic expression may include logical expression with the following convention: 
 
 A TRUE expression takes the value 1 
 A FALSE expression takes the value 0 
 
Under this convention the arithmetic expression: 
 
 (POP99.AGE99 > 15)  
 
takes  the value 1 for any person of 16 years old and more and the value 0 for persons less than 16 years 
old. 
 

 
Demonstration 

Classify the households according to its type of Floor material: satisfactory and unsatisfactory. 
 
RUNDEF CMD23 
   SELECTION ALL 
 
DEFINE HOUSE99.UBNfloor 
AS  (HOUSE99.FLOOR99 = 4) 
TYPE INTEGER 
RANGE 0 - 1 
 
TABLE one 
AS FREQUENCY 
OF HOUSE99.UBNfloor 
 
 

 
Exercises 

1. Classify the household according to the energy supply into two categories: 0 "satisfactory" and 1 
"unsatisfactory". 

2. Classify the household according to the type of sewer into two categories: 0 "satisfactory" and 1 
"unsatisfactory". 

3. Classify the household into two categories depending to the water source: 0 "satisfactory" and 1 
"unsatisfactory". 

QUIZ 3 
 
Name: ________________________________________ 
 
Date: _________________________________________ 
 
 
Please answer the following questions 
 
1. When and why do we use the RECODE clause? 
__________________________________________________________________ 
__________________________________________________________________ 
__________________________________________________________________ 
__________________________________________________________________ 
 
2. Identify the generations inside a household using the relation to household head using the following 
scheme: 
 

 
Generation of   Recode value 

Household head & spouse   1 
Children  (son)    2 
Grandchildren (father plus son)  3 
Parents (father & mother)   4 
Non identifiable (other)   6 
 
 
 
3. When and why do we use the SWITCH  clause? 
__________________________________________________________________ 
__________________________________________________________________ 
__________________________________________________________________ 
__________________________________________________________________ 
__________________________________________________________________ 
 
4. Write down the command set using switch with the following requirement: Assign a 1 to young females 
(10 – 17) attending school, assign a 2 to young females (10 – 17) not attending school, assign a 3 to young 
males (10 – 17)  attending school, assign a 4 to young males (10 – 17) not attending school. 
__________________________________________________________________ 
__________________________________________________________________ 
__________________________________________________________________ 
__________________________________________________________________ 
__________________________________________________________________ 
__________________________________________________________________ 
__________________________________________________________________ 
__________________________________________________________________ 
__________________________________________________________________ 
__________________________________________________________________ 

XII. Creating new indicators 
 

Creating a new variable that defines de household Type 
(HHOLDTYPE) 
 

A household type category is a way to differentiate their inner household family structure. It is a computed 
variable that has being processed with census variables extracted from the database. 

Definition 

 

 
Categories 

1. Head of the household and spouse 
2. Single headed household male without children  
3. Single headed household female without children  
4. Single female headed household with children < 15 
5. Single male headed household with children < 15 
 

Notes 
The use of the variable helps to segment or stratify the population into groups. One of these groups is the 
“Single headed female household with children” which might be used as a pivot to compare basic household 
statistics such as water and toilet conditions, electricity supply, education level and any other variable that 
can provide insight on the population socio economic profile classified by family structure. 

Promoting information – Selecting a unique record 
 

The objective of this exercise is to demonstrate the technique for creating a new entity variable (age of 
household head) from a member entity variable (age of individual). This operation “promotes” a value from a 
lower level of the hierarchy. In various situations, a specific record of a lower entity may be promoted to its 
parent entity without loosing any relevance. This is the case when the record has a unique characteristic 
within the entity children.  

Objective 

 
The traditional case is the individual of household selected as the household head. In this case, any 
attribute of  the person (age, sex, etc.) characterized the household itself. 
 
Based on this statement, a household can be characterized by the sex of its head. The variable at person 
level can be “promoted” to household level. The scheme is implemented with a FOR clause. 
 
Expressed in natural language, the predicate for realising this operation would be
Define the “sex of household head” as the “sex of the person” who is “identified as head of 
household”. 

: 

 
The translation of this predicate into a more REDATAM+SP structure would be
Define at the “household entity level” a new variable called “sex of household” 

: 

as the “sex of the individual”  considered  “the head of household”. 
 
The translation of this predicate into the exact REDATAM+SP syntax: Defining ages and sex of a household 
head: 
 DEFINE HOUSE99.SEX 
  AS POP99.SEX99 
  FOR POP99.RELATION99 = 1 
   
 DEFINE HOUSE99.AGE 
  AS POP99.AGE99 
  FOR POP99.RELATION99 = 1 
   
DEFINE indicates that a new variable is to be created. “HOUSE99.SEX” is the syntax's convention to define 
the variable identifier. The first part of the identifier is the entity name to which belongs the variable, then a 
dot then the variable name. This name is user defined. The variable “SEX” belongs to, or “is member of”, the 
entity “HOUSE99”. The AS clause describes what value will take the new variable and the FOR clause is 
used to select the individual record. The LIKE clause is a short cut to assign all attributes (type, range, size, 
etc.) of a variable to the one currently defined. 
 

1. The scheme is relevant when there is one and only one person defined as the head of household. Unless 
the data contains some errors this assumption is reasonable.  

Notes 

 
2. Whenever the previous assumption is FALSE the variable is considered MISSING. To prevent this 
situation a default value must be defined. 

Deriving poverty indicators from the census Data  
 

This module is focused on substantive matters more than in techniques. The selection of the appropriate 
variables is important to generate a profile of the households through various indicators. The Unmet basic 
needs (UBN) indicator measures the physical conditions under which population lives as well as their level 
of economic vulnerability.  

Objective 

 
Unmet Basic Needs 

• Physical : Dwelling state 
• Services and Utilities : Water, Electricity, Sewage 

 
Vulnerability 

• Household Head Education Level 
• Dependency 

 
 
A. Defining basic needs related to physical state of dwelling 
 
Produce tables related to physical variables included in the household questionnaire. Comparison is made 
according to urban versus rural location and the year of building. 
 
*Type of dwelling 

TABLE table1  
   AS CROSSTABS 
   OF HOUSE99.SEWER99  BY DIST99.DISN99 

*Ownership 
TABLE table2 
   AS CROSSTABS 
   OF HOUSE99.WATER99 BY DIST99.DISN99 

*Wall material 
TABLE table3 
   AS CROSSTABS 
   OF HOUSE99.WALL99 BY DIST99.DISN99 

*Roof material 
TABLE table4 
   AS CROSSTABS 
   OF HOUSE99.ROOF99 BY DIST99.DISN99 
 

 
Note: 

The AS clause is a mix between arithmetic and logical expression. A logical expression is always of type 
BOOLEAN (true, false). A BOOLEAN variable is NOT an INTEGER variable. However, a BOOLEAN value 
True is coded as a 1 value and a False value is coded 0. The AS clause uses this unorthodox feature. The 
SWITCH command can be used also. 

Defining a Sex & Dependency ratio 
 
Among the traditional demographic indicators, the sex ratio and dependency ratios occupy the top of the list. 
REDATAM+SP offers a very simple way to compute these indicators through the use command sets.  
 

 
Demonstration 

Vulnerability is defined as a household indicator. Economic dependency is defined by measuring the ratio 
between the economically active and economically inactive persons. This ratio is then  
 
a) Vulnerability is defined as a household indicator. Economic dependency is defined by measuring the ratio 
between the economically active and economically inactive persons. This ratio is then active people divided 
by the number of inactive people. The definition of active and inactive depends on the user definition and is 
generally calculated from an Economic Activity variable of the database.  
 
 
 

 
Computation scheme 

RUNDEF IND1 
    SELECTION all 
 
DEFINE POP99.active 
    AS RECODE POP99.91actlwk (1, 2 = 1) ELSE 0 
    TYPE INTEGER 
 
DEFINE ed.active 
    AS COUNT POP99 
    FOR POP99.active = 1 
    TYPE INTEGER 
 
DEFINE ed.allpersons 
    AS COUNT POP99 
    TYPE INTEGER 
 
DEFINE ed.indic 
    AS (ed.allpersons - ed.active) / ed.active 
    FOR ed.active > 0 
    TYPE REAL 
 
TABLE indicator 
    AS AREALIST  
    OF ed, ed.indic  

 
 
 

 
Notes 

1. Dividing by zero. The FOR clause prevent a program abort if, the number of actives in an area is equal to 
zero. In this case, the dependency ratio for the area is considered NOT APPLICABLE 
 
 
b) The sex ratio of a given area is defined as the ratio of the number of males divided by the number of 
females in the area. The index is generally computed as a percentage from 0 to 100.  

Command set  

RUNDEF IND4 
    SELECTION ALL 
 
DEFINE distr.totmales 
    AS COUNT POP99 
    FOR POP99.SEX99 = 1 
    TYPE INTEGER 
 
DEFINE distr.totfemales 
    AS COUNT POP99 
    FOR POP99.SEX99 = 2 
    TYPE INTEGER 
 
DEFINE distr.indic 
    AS 100 * ( distr.totmales / distr.totfemales ) 
    FOR distr.totfemales > 0 
    TYPE REAL 
 
TABLE distr.ratio 
    AS AREALIST  
    OF distr, distr.indic   
     

 
 
 

 
Exercise 

1. Create a thematic map of the dependency ratio at the ed level. 
2. Compute a sex ratio for population between 15 and 45. 
3.  Create a thematic map for this indicator. 

Estimating the education Level of the Head of the Household 
 
 
Define for each household the level of education given by the education attained by the Head of the 
household. Define what you will considerer low level medium or high level of education and then assigned it 
to each person. Then pick the head of the household and promote it to the household itself. 
 
 

 
Command set 

RUNDEF IND4 
    SELECTION all 
    UNIVERSE POP99.RELATION99 = 1 
 
DEFINE POP99.var 
    AS RECODE POP99.91edlvl ( 1, 2, 3 = 1 ) ELSE 0 
    FOR POP99.AGE99 >= 15 
    TYPE INTEGER 
 
DEFINE ed.selection 
    AS COUNT POP99 
    FOR POP99.var = 1 
    TYPE INTEGER 
 
DEFINE ed.totrecords 
    AS COUNT POP99 
    FOR POP99.AGE99 >= 15 
    TYPE INTEGER 
 
DEFINE ed.indic 
    AS ( 100 * ed.selection)  / ed.totrecords  
    FOR ed.totrecords > 0 
    TYPE REAL 
 
TABLE indicator 
    AS AREALIST  
    OF ed, ed.@ED, ed.indic  

Poverty indicator:  

Socio-economic status of teenage mothers. Is poverty a 
determinant of teenage fertility? 
 

 
Objectives: 

• Acquire the capability of translate a conceptual problem into data requirements and command sets 
to execute the statistical processing 

• Demonstrate the step by step analysis of a complex statistical process. 
 

Definition of a target group. Cross tabulation with the target group. Analysis of differences between total 
population and specific target group. Example using  a sample data base. 

Strategy: 

 
 

Step 1 - Target group definition 
 

 
Group definition 

Our population to target involves women up to 19 years old that have live children. 
 
To find these women we have several variables: age (AGE99), sex (SEX99), fertility filter (91fert) and live 
births (91livbth). Using these variables (not necessarily all of them) we classify the persons that belong to 
this group saving the classification in a new variable (91targetgr). The command set would look like the 
following: 

 
RUNDEF Target 
   SELECTION ALL 
DEFINE POP99.targetgr 
    AS (POP99.91livbth > 0  AND POP99.AGE99 < 20) * 1 + 
       (POP99.91livbth = 0  AND POP99.AGE99 < 20) * 2 
    RANGE 1-2 
    OPTIONS DEFAULT 0  NOTAPPLICABLE 9 
    TYPE INTEGER 
    VARLABEL  "Women < 20 Target Group"  
    VALUELABELS 1 "with kids" 2 "no kids"  
    SAVE targetgr.rbf 

Step 2 - Household conditions using RECODE 
 
The following step involves the creation of a more complex indicator that classifies the dwelling according to 
its conditions. This definition was done in previous exercises so now we are going to modify some 
parameters. 
 
Analyze the following REDATAM+SP command set:  

RUNDEF IND2 
    SELECTION all 
DEFINE HOUSE99.var1 
    AS RECODE HOUSE99.91shrdt (2 = 1 ) ELSE 0  
    RANGE 0 - 1  
    TYPE INTEGER 
DEFINE HOUSE99.var3 
    AS RECODE HOUSE99.91water (4,5,6,7 = 1 ) ELSE 0  
    RANGE 0 - 1  
    TYPE INTEGER 
DEFINE HOUSE99.var4 
    AS RECODE HOUSE99.91roof (7,8 = 1 ) ELSE 0  
    RANGE 0 - 1  
    TYPE INTEGER 
DEFINE HOUSE99.var5 
    AS RECODE HOUSE99.91light (1,2,4 = 1 ) ELSE 0  
    RANGE 0 - 1  
    TYPE INTEGER 
DEFINE HOUSE99.var6 
    AS RECODE HOUSE99.91wall (6,7,8 = 1 ) ELSE 0  
    RANGE 0 - 1  
    TYPE INTEGER 
DEFINE HOUSE99.var7 
    AS RECODE HOUSE99.91toilet (3,4,5 = 1 ) ELSE 0  
    RANGE 0 - 1  
    TYPE INTEGER 
DEFINE HOUSE99.var8 
    AS RECODE HOUSE99.91kitch (2 = 1 ) ELSE 0  
    RANGE 0 - 1  
    TYPE INTEGER 
DEFINE HOUSE99.var 
    AS HOUSE99.var1 + HOUSE99.var3 + HOUSE99.var4 +  
       HOUSE99.var5 + HOUSE99.var6 + HOUSE99.var7 + HOUSE99.var8 
    RANGE 0-8 
    TYPE INTEGER 
    VARLABEL  "Households with UBN"  
    VALUELABELS      0 "none" 1 "1 UBN" 2 "2 UBN" 3 "3 UBN" 4 "4 UBN" 
                                   5 "5 UBN" 6 "6 UBN" 7 "7 UBN" 8 "8 UBN" 
DEFINE ed.indic 
    AS COUNT HOUSE99 
    FOR HOUSE99.var >= 3 
   VARLABEL  "Number of Households w/more 3 ubn by ED" 
    TYPE INTEGER 
 
TABLE ed.ratio 
    AS AREALIST  
    OF ed, ed.indic 

Now, it is necessary to modify this indicator to classify each household instead of just counting them for 
each ED. To do this modify the last DEFINE 
 

DEFINE HOUSE99.indic 
    AS HOUSE99.var >= 3 
    RANGE 0-1 
    VARLABEL  "Household w/more 3 ubn"  
    VALUELABELS           0  "< 3 UBN"         1  "> 3 UBN"  
    TYPE INTEGER 
 
 

Step 3 - Household dependency ratio 
 
Dependency ratio at household level.  
 
The dependency measures the burden of each active person of the house who has to maintain those 
inactive persons. The active members in the household can be obtain by the variable of Activity last Week 
(91actlwrk) where we find the persons that worked, look for a work, studied, etc. during the previous week of 
the census. Form this variable we get the actives members and from the count of all the members in the 
household we obtain the indicator. There are two ways of computing a dependency ratio: a) where we 
measure the percentage of active members in the household, and b) where we obtain the number of 
dependents by each active member.  

 
a)  computes % of active persons in the household 
DEFINE HOUSE99.depratio 
  AS ( 100 * HOUSE99.actives / HOUSE99.totpers ) 
  FOR HOUSE99.totpers > 0 
  TYPE INTEGER 
  RANGE 0-100 
 
b) computes the number of dependents by each active person in the household 
DEFINE HOUSE99.indic 
    AS (HOUSE99.totpers - HOUSE99.actives) / HOUSE99.actives 
    FOR HOUSE99.actives > 0 
    TYPE REAL 
    OPTIONS DEFAULT 0 

Adding a threshold or a break point to this count allows us to classify the household into good or bad.  
 
Assignment
 

: 

Classify those households with more than or less than 30 % of active members. What is the distribution of 
the dependency ratio? 
 
Compare this indicator with our target group. What is the percentage of teenage mothers that live in 
households with less than 30 % of active members? 

 

Step 4 – Education Dimension 
 
Education is measured in several variables so we have to define which ones are going to be analyzed. For 
example: School attending mode (EDREACH99), attending school (SCHOOL99), Education level 
(EDUCATTN99). 

 

 
Assignment: 

1. Calculate the percentages of  teenage mothers (less 20 years old)  that only reached primary level. 
2. From the relation between teenage mothers and their education level, what can be said? Is there any 

difference among the rest of the population? 
3. Calculate which of these teenagers gave birth. Has the burden of childbearing limited their education? 
 

Step 5 – Working situation 
 
Assignment
 

: 

1. Analyze the economic the relation between our target group and the economic activity. 
2. Focus on the total population and extract the head of households characteristics such as education 

level, economic activity, unemployment, etc. and analyze the distribution. 
3. From here get the female head of households only.  
4. Determine the age of the head of the household to compare minor and adults head of households with 

the unemployment variable.  
5. Crosstabs between Household Headed by a minor by Household Head Sex.  
 

Step 6 - Indicator summary 
 
After you have analyze the different dimension of teenage mothers including their family structure, 
education, level, economic activity, vulnerability and poverty. You can summarize the analysis presenting a 
full report with tables outputs, and maps.  

Matrix Operations example: 
 
The MATRIX operation is a function that operates over computed tables that have the same dimension 
(same number of rows and columns). The operations are very similar to the ones of common matrices 
(multiplication, division, sum, subtractions). It is very useful when you need to get percentages and ratios of 
values ordered in table format.  
 
 
EXAMPLE: Compute a dependency ratio based on age 
 
RUNDEF job 
   SELECTION ALL 
 
***1. CREATE THE VARIABLE AREA AT THE PERSON LEVEL ************ 
 
DEFINE PERSON.AREA 
   AS  VAL (AREA.AREA) 
   RANGE     1 - 2 
   TYPE INTEGER 
   VALUELABELS 1 "URBAN" 2 "RURAL" 
    
******2.  CROSSTABULATION OF URBAN-RURAL BY SEX BUT JUST FOR PEOPLE LESS THAN 14 AND  
*** OVER 65 YEARS OLD ***************** 
 
TABLE P1 
TITLE "PEOPLE LESS THEN 14 AND OVER 65" 
AS CROSSTABS OF PERSON.AREA BY PERSON.SEX 
 FOR PERSON.AGE <=14 OR PERSON.AGE >= 65 
 
**** 3. CROSSTABULATION OF URBAN-RURAL BY SEX BUT JUST FOR PEOPLE BETWEEN 14 AND  
**** 65 YEARS OLD ********** 
 
TABLE P2 
TITLE "PEOPLE BETWEEN 14 AND 65" 
AS CROSSTABS OF PERSON.AREA BY PERSON.SEX 
 FOR PERSON.AGE >= 15 AND PERSON.AGE <= 64 
 
****** 4. TO GET THE DEPENDENCY INDICATOR (BASED ON AGE) YOU DIVIDE TABLE 1 BY TABLE 2 
 
 TABLE P3 AS MATRIXOP OF P1, DIVISION, P2 
 
 *** 5. MULTIPLY THE OUTPUT BY 100 TO GET THE PERCENTAGE **** 
 **** IT MEANS HOW MANY CHILDREN AND ELDERLY DEPEND ON THE REST OF PEOPLE 
 
TABLE AR010103P 
TITLE "DEPENDENCY RATIO BASED ON AGE" 
        AS MATRIXOP 
        OF P3, MULTIPLICATION, 100 
    OPTIONS       DECIMALS 1 
        OUTPUTFILE XLS "C:\Bases\INDCELADE\Indics_proyecto BID\AR010103_PAIS.XLS" 
            OVERWRITE 

Defining a map composition 
 
This R+SP version allows you to display thematic maps of database variables that are associated with a 
geographical code. This feature is very important for planning, since the maps show at first site which areas 
have a similar value, and which areas are different, according to the mapped variable. By using different 
cartographic coverages you can see how the geographic objects such as rivers, mountains, etc., can affect 
the behavior of the variables being displayed. (For example, the access to schools can be relevant when 
explaining different literacy levels). 
 
Working with maps in R+SP depends upon the existence of such maps in a Shape format of ArcView™. 
R+SP system DOES NOT builds digitized maps, it only converts the SHAPE format to an internal R+SP 
working format. 
Once the database is opened, the procedure to connect a map to the base is executed by calling the Map 
Composition function. 
 

I. PROCEDURE 
 
In the main menu, choose the option File >New >Map Composition. 
 
The system will sow the Map Properties dialog box. In the General panel, press the Add button. R+SP will 
open a new dialog box to allow you to choose a Shape file (generated by ArcView). Such file must contain 
the same entity elements you want to display in the database. 

 
 
Select the *.shp file and press the Open button to accept it. If you want to modify an existing map 
composition, use the File >Open >Map Composition in the main menu, and select the corresponding map 
file (.mxp extension). 

Exercise 

1. Create a map composition with districts. 
 
 

II. THE PROPERTIES WINDOW 
 

General 
 
First of all, in the General panel you must define the name for this map composition, and enter it in the Map 
Name box. Then, press the Add button to associate a coverage to this map composition that you are 
creating. The map composition can have as many coverages as you need (for example, the administrative 
map, the hydrological map, etc.) 
The Delete button can be used to erase a coverage. 
The Add button displays another dialog box for you to choose the shape file (.shp) containing the coverage. 
Once accepting the coverage file the system will display the map on the screen. 
 

 
 

 
Coverage 

If you want to keep editing the map composition properties, you should select the Coverages panel in the 
map properties window. 

This window contains three sub-panels, General, Links and Default Symbol, as you can see in the 
following picture. 
 

 
 
The Links sub-panel is the most important, it is used to associate the map with the database. In order to 
do that the system shows a combo box with all the variables in the map file. You should select the field from 
the map that holds the key field (Redatam code) associated with the Redatam database (related to the 
entity level that the map is reopresenting, ex. A district map is related with the Redatam base with a district 
code). Remember that the code in Redatam is a compound code. Select the entity level in the lower combo 
box.  
 
The map variable name is selected in the Map Code combo box, and the database entity is selected in the 
R+SP Entity combo box. 
 
Just for you to know, shape files have associated dbf files containing fields with information about the map. 
If you want to join this map with other files, one of such fields must have the same characteristics and 
values as the entity (or field) in the database that you want it to be connected with. 
 

Presentation 
 
The Presentation panel of the Map Properties window has six subpanels: Page, Title, SubTitle, Comments, 
Map and Legend, as you can see in the picture below. 

Classification 
 
The Classifications panel of the Map Properties window allows you to define the range types that will be 
used to represent the database variables in the map. 
The Classification Scheme defines the classification criteria used to sort the geographical entities. They 
can be: 
 

No Classifications No Classification will be used. 
Quantiles (5 or 3 groups) Define entity groups (5 or 3), each group with the same number of entities. 
Intervals (5 or 3 groups) Define entity groups (5 or 3), each group with the same range size. 
Unique Values Single values for each entity element. 

 
You can define also the colour for the minimum and maximum categories (groups), and the colours for the 
not applicable and missing values. 

Page 
 
This property allows you to define a border for the map, using the Border option.  
You can define the background colour for the map, using the Background Colour property, and select the 
default Font property for the map texts. 
 

Title 
 
This panel allows you to insert a title in the Map Composition, and to define the title characteristics: the 
Visible and Border boxes are to define if the title will be visible and if it will have a border. The Top and 
Left boxes define the title position referring to the page. The Text box is for the default title. You can also 
define other title properties like size, bold, underline and italics. 
 

Subtitle 
 
This panel has the same options as the Title one: Visible, Border, Top, Left, Text, size, bold, underline and 
italics. It is used to insert a subtitle into the Map Composition. 

Comments 
 
To insert a comment into the Map Composition, with the same options of the Title and SubTitle panels. 

Map 
 
This panel defines the frame properties that will contain the map in the Map Composition. It has the 
following fields: 

Left Distance from the left margin of the page. 
With Frame width. 
Top Distance from the top margin of the page. 
Height Frame height. 
Colour Define the frame background colour, or define a transparent background. 

 
There are also the following adjustment buttons: 

Adjust Frame Automatically adjust the frame size according to the map size. 
Adjust Map Automatically adjust the map size according to the frame size. 

 
The Map panel properties can be seen in the following picture. 
 

 
 

 
Legend 

This panel defines the legend presentation in the map page, containing the categories of the variable you 
are mapping. The options are: Visible, Title, Font size, Borders (None, Simple, Double), the border width 
and the line width (Line, Border), and the legend’s background colour. 

Save the Map Composition Properties 
 
Once you have defined all the properties of the Map Composition, you should save it by using the Save 
button in the properties window. It will be saved in a text file with an .mxp extension, and this file can be 
edited with any text editor if you want to change its contents. 
 
This means that there are two ways to change the Map Composition properties. One is by using the R+SP 
internal tools that you can find in the menus; the other one is directly editing the mxp file manually. 

III. DISPLAYING INDICATORS 
 
Once an indicator has been processed, the results are displayed in the output window at the same time if 
the map composition is open the new variable or indicator is listed in the combo box in the middle of the 
Map Window Toolbar. 
 
 

 
 
In the map above, the user selected and processed an AREALIST of the variable “Toilet facilities” at the 
district level. Other map properties such as the Title, Subtitle, Legend, Coverage and Classifications where 
defined in the map properties window. 
 
 

 
Exercise 

1. Create a map composition at the Province  level. Change the properties of the Title and subtitle. 

